FINESSEの使い方マニュアル

宗宮 健太郎*

平成 25 年 6 月 19 日

1 はじめに

FINESSE は Andreas Freise 氏が作ったシミュレーションコードで、干渉計の信号出力を周波数領域で数値計算することができる。同様のコードである Optickle と比べると、FINESSE は輻射圧を含めた計算ができない一方で、空間高次モードを計算できる強みがあり、用途によって両者を使い分けるのがよいだろう。また、Optickle は Matlab がないと動かないが、FINESSE は完全に無料のソフトである。まず Andreas の homepage に行き、FINESSE を download することから始める。

http://www.gwoptics.org/finesse/

Homepage には Linux 用や Mac 用もあるが、ここでは Windows 用をダウンロードしたとして話を進める。ダウンロードして ZIP ファイルを解凍すると、実行ファイルである KAT.EXE と設定ファイルである KAT.INI が入っている。上記の homepage には Andreas の書いた英語のマニュアルも置いてある。

FINESSE を使うには、gnuplot というフリーのグラフソフトもインストールしておくと便利である。Gnuplot をダウンロードして、そのフォルダを C ドライブの直下に移動する。FINESSE から gnuplot C path を通すために、KAT.INI を editor で開き、

"C:/gnuplot/bin/wgnuplot.exe"

などと書く (類似の記述がある行があるので、そこを書き換える)。ここに書き込むのは gnuplot の実効ファイルが置いてある場所であり、さきほどダウンロードした gnuplot フォルダの中のどこに wgnuplot.exe があるかを探し、上記と違う場合は訂正する。

2 例 1:マイケルソン干渉計

では、まず拡張子を.kat としたファイルを作り、以下のコマンドを書き込む。ファイル名は仮に test.kat とする。

- (01) l i1 100 0 nL
- (02) s sEOM 1 nL nEOin
- (03) mod eo1 10M 0.1 2 pm 0 nEOin nEOout
- (04) s sRE 1 nEOout nPOin
- (05) bs po 0 1 0 45 nPOin dump nPOout npo
- (06) s sBS 1 nPOout n1
- (07) bs bs1 0.5 0.5 0 45 n1 n2 n3 n4

 $^{{\}rm *Corresponding\ email: somiya@phys.titech.ac.jp}$

- (08) s sAS 1 n4 nAS
- (09) s north-s 3000 n2 n5
- (10) s east-s 3001 n3 n6
- (11) m north-m 0.99995 50e-6 0 n5 n7
- (12) m east-m 0.99995 50e-6 90 n6 n8
- (13) pd1 Ctrl 10M 90 nAS
- (14) xaxis north-m phi lin -180 180 100
- (15) func y = 90- x1
- (16) put* east-m phi \$y
- (17) noplot y
- (18) yaxis lin abs
- (19) pause

説明は後回しにして、まずこれが計算できるかどうか確認してみよう。このファイルをどこかのフォルダにいれて、そのフォルダを指定してコマンドプロンプトを開く。具体的には、プログラム アクセサリ コマンドプロンプトと選択してコマンドプロンプトを開き、

C: Y Users Y hoge >

とあるので、これより上のフォルダに行く場合は cd.. と打ち、下のフォルダに行く場合は cd hogehoge などと打ち込む。この作業が面倒ならば、いじくるつくーるというソフトをダウンロードすると、右 クリックでコマンドプロンプトをその場で開くことが可能となる。

さて、kat ファイルの置いてあるフォルダを指定したコマンドプロンプトで

kat test1

と打ち込むと FINESSE が実行され、図1 が表示される。また、test1.out が作成され、その中には掃引位相と出力ともう1 つの合計3 列が記録されている。




図 1: サンプルコードを実行したときの出力

ちなみに、最近アップデートされた Windows 版の FINESSE を走らせると gnuplot が表示されないようである。その場合は私に相談してください。

3 KATファイルの中身

それでは、サンプルコードで用いられているコマンドを順番に見ていこう。(01) 行目から(12) 行目には干渉計構成が記載されている。構成要素ごとに使用方法を以下にまとめる。

3.1 Mirror

Mirror はm で表し、以下のようなパラメータを書き込む。

m 名前 反射率 透過率 位相シフト ノード1 ノード2

ここで、FINESSE は元々cavity などの長さを carrier に対して波長の整数倍に設定するので、位相シフトという項はそれを反共振や非共振に変更するためのものである。ノードというのは KAT 内の絵に n1 などと描いてあるように、場所を表すものである。その先に何もないときは dump と書けばよい。

3.2 Space

Mirror 間の距離はsで表す。

s 名前 距離 (m) ノード ノード

ここで、光速 c は KAT.INI に 7 桁で書いてあるから、距離もそれに合わせて 7 桁くらいまで計算して書き込まないといけないので注意。

3.3 Beam Splitter

BS は bs で表す。

bs 名前 反射率 透過率 位相シフト 入射角 ノード ノード ノード ノード 入射角の単位は度。ノードの順番は入射、反射、透過、それ以外の順番となっている。

3.4 LASER

LASER は1で表す。

1 名前 強度 (W) 周波数 ノード

周波数は KAT.INI に書いてあるデフォルト値とのズレで、たいてい0でよい。

3.5 変調器

変調器は mod で表す。

mod 名前 変調周波数 変調指数 次数 種類 ノード ノード

変調周波数は $10^7(\mathrm{Hz})$ とかは $10\mathrm{M}$ と書いてよい。次数のところで、single sideband にしたければ、s と書く。種類のところは、位相変調なら pm 、強度変調なら am と書き込む。

4 誤差信号の計算

まずフォトデテクタを設置する。通常の復調では、pd1 名前 復調周波数 復調位相 ノード と書く。フォトデテクタは 40 個まで設置が可能で、複数置くとそれらの出力が同時に表示される。サイドバンド同士のビートをとる二重復調をする場合は、

pd2 名前 復調周波数 復調位相 復調周波数 復調位相 ノードを用いる。

掃引は以下のコマンドを使って行なう。

xaxis 振る物 どのパラメータを振るか lin 掃引始点 (deg) 掃引終点 (deg) 点数 vaxis abs

掃引幅を小さくして点数を 1 にすれば表示値が出力の微分幅を表す。そして yaxis コマンドは abs に すると出力が gain に、deg にすると出力が phase になり、abs:deg にすると同時に表示される (cavity を含まないと phase は表示されないらしい)。

さらに、2 つの mirror を同時に掃引するときは以下のコマンドを xaxis コマンドと yaxis コマンド の間に書き込む。まず、新しい関数を設定し、

func 関数名 = 数式

そのあとで

put 振る物 どのパラメータを振るか \$ 関数名

と書く。数式の中ではxaxisで与えられる値をx1で表す。サンプルファイルでは、east-mを north-m と反対向きに掃引して差動信号を見たいので、位相を 90 度ずらしている (往復で 180 度)。

ここでデフォルト位置というものが重要である。Mirrorを設定するときに位相シフトを書き込んだが、掃引対象に関しては一度リセットされてしまう。ここで put*とアスタリスクをつけると定義したデフォルト位置を保存することができる。サンプルコードでアスタリスクを外すと波の位置がずれるのを確認できる。

5 伝達関数の計算

伝達関数は、RF 変調に加えて信号サイドバンドを用意し、二重復調することで計算する。サンプルコードの (13) 行目以降を以下のように書き換える。

- (13) fsig sig1 east-m 1000 0
- (14) fsig sig2 north-m 1000 180
- (15) pd2 GW 10M 0 1000 max nAS
- (16) xaxis sig1 f log 1 10k 100
- (17) put GW f2 \$x1
- (18) scale meter
- (19) yaxis log abs
- (13)(14) 行目で fsig コマンドを用いて信号サイドバンドをたてる。最後から 2 番目の数字が信号サイドバンドの周波数である。ここでは 1kHz に設定しているが、実はこの数字はなんでもよい。(15) 行目で RF サイドバンドと信号サイドバンドで二重復調を実施する。信号サイドバンドの復調位相に関しては \max コマンドを用いる。RF サイドバンドでは \max コマンドは利用できないようである。(16) 行目で信号サイドバンドの周波数を掃引する。ここで sig1 の周波数を掃引すれば sig2 も同時に掃引さ

れる。(17) 行目で ${f put}$ コマンドを用いて、二重復調の周波数も合わせて掃引する。(18) 行目は単位をデフォルトの $W/{f rad}$ から $W/{f meter}$ に変えるものである。

6 ショットノイズの計算

ショットノイズの計算は、伝達関数の計算に用いたフォトデテクタを pdS2 に変えるだけでできる。 (15)~pdS2~GW~10M~0~1000~max~nAS

このとき (18) 行目で単位を変えてあることが重要で、それにより単位が $\mathrm{m}/\sqrt{\mathrm{Hz}}$ で表されることになる。